SYSTEMS ELECTRONICS GROUP

SYSTEMS M4500 INDUSTRIAL CONTROLLER

S4580: ANALOG I/O BOARD
4 ANALOG INPUTS AND 4 ANALOG OUTPUTS

- **4 Single Ended Analog Input Points and 4 Single Ended Analog Output Points**
- **14-Bit, Bipolar Analog-To-Digital Conversion on Input Points**
- **12-Bit, Bipolar Digital-To-Analog Conversion on Output Points**
- **Input Range: -10V To +10V**
- **Output Range: -10V To +10V**
- **Fast Conversion Times: 15 Microseconds per Input and 10 Microseconds per Output**
- **Flexible I/O Update to Facilitate a Wide Variety of Applications**
- **Removable Field Wiring Connector**
- **Standard M4500 I/O Form Factor**



General Description

The S4580 Analog I/O board contains 4 analog inputs and 4 analog outputs on one single width board. The analog inputs are converted to digital values which can then be read by the M4500 processor. The M4500 processor writes digital values to the S4580 for each output point which are then converted to the corresponding analog voltage.

The input section consists of four single ended analog inputs which are multiplexed to one analog-to-digital converter (ADC). The ADC converts the analog value for each input, one at a time, to the corresponding digital value. The ADC is 14-bit, bipolar successive approximation converter with a full range of 0 to 8191 corresponding to 0 to +10V and 16383 to 8192 corresponding to -0 to -10V respectively. The standard input voltage range is -10V to +10V resulting in a resolution of 1.22mV per bit. The conversion time of the ADC (including input select and settling time) is 15 microseconds.

General Description (cont'd)

The output section consists of four single ended analog outputs, each equipped with it's own digital-to-analog converter (DAC). Each DAC converts the corresponding digital value for the output to an analog voltage representing the digital value. The DACs are 12-bit, bipolar digital-to-analog converters with a full range of 800H to fffH corresponding to 0 to +10V and 0H to 7ffH corresponding to -10V to -0V respectively. The standard output voltage range is -10V to +10V resulting in a resolution of 4.88mV per bit. The conversion time of the DACs is 10 microseconds.

Power for the analog section (+/-12V) is supplied via the backplane on M4510, M4512, and M4513 chassis. For all other M4500 modules/chassis, +/-12V power must be supplied via the removable field wiring connector. Refer to figure 2 for typical I/O field wiring connections and +/-12V power wiring.

Program Interface

As with other M4500 I/O boards, the I/O update of the S4580 is not done automatically by the M4500 processor. This provides maximum flexibility for I/O update by allowing the user to implement the I/O update as necessary (in the main program, timed interrupt, or input interrupt, etc. as required by the application). The S4580 I/O update is more complicated than standard I/O update since the user is now directly interfacing with the ADC and DACs and thus has to perform functions such as selecting the desired analog input, initiate the conversion, wait for the conversion, read the input data, etc..

The S4580 contains a "Device Select" register which is used to select the device (ADC, DAC, Analog select, etc.) that will be accessed. This register must always be set prior to accessing the desired device, otherwise erroneous results will occur. The "Device Select" register is write only and is written by setting PORT2.7 to a "1" and then writing the select command to byte 0 of the S4580 slot address.

The following command codes select the respective devices on the S4580:

Command Code	Device Selected
00H	Analog Input Select
01H	ADC Select
02H	Conversion Initiate
03H	DAC0/1 Select
04H	DAC2/3 Select

In addition to the above device select codes, the control bit PORT2.7 is used to select either the "Device Select" register or to actually access the respective device specified in the "Device Select" register. When PORT2.7 is "1", the "Device Select" register is accessed. When PORT2.7 is "0", the respective device specified in the "Device Select" register is accessed.

Refer to the following example program for details on reading the analog inputs and writing the analog outputs.

Program Example

The following is an example of both reading analog inputs and writing analog outputs. This program uses two user functions: ufunc02() to read a specified analog input and ufunc04() to write all the analog outputs. These two user functions are called from the main program, thus the analog I/O update would occur in the main program scan.

Main Program:

```
*****************************************************************************************************************************
block:  1 - High level

W104 = 7424;                    /* S4520 slot address (S4580 located in slot 0-0 */
B102 = 0;                       /* Analog Input  to read */
ufunc02();                      /* read Analog Input0 */
B102 = 1;                       /* Analog Input to read */
ufunc02();                      /* read Analog Input 1 */
ufunc04();                      /* write all Analog Outputs */
```

User function 2:

```
*****************************************************************************************************************************
block:  1 - Assembly

; This user function reads one analog input as selected by “ASEL”. Prior to calling this user function,
; “ASEL” must be loaded with the analog input to be read (either 0,1,2, or 3) and “BADDR” must be
; loaded with the byte 0 slot address the S4580 is in (i.e. 7424 for slot 0-0, 7488 for slot 0-1, etc.).
; The user function will return with the analog value in “AIN” (W100) [0-8191 corresponds to 0 to +10V,
; 16383-8192 corresponds to -0 to -10V respectively].

                .equ    AINL,B100
                .equ    AINH,B101
                .equ    ASEL,B102
                .equ    ADVCE,B103
                .equ    BADDR,W104

; select Analog Input
                orb     PORT2,#80h          ;set “Device Select”
                ldb     ADVCE,ZERO_REG      ;select “AIN Select”
                stb     ADVCE,[BADDR]       ;write to “Device Select” register on S4580
                andb    PORT2,#7fh          ;reset “Device Select”
                stb     ASEL,[BADDR]        ;select Analog Input to read

; initiate ADC Conversion
                orb     PORT2,#80h          ;set “Device Select”
                ldb     ADVCE,#02h          ;select “Conversion Initiate”
                stb     ADVCE,[BADDR]       ;write to “Device Select” register on S4580
                andb    PORT2,#7fh          ;reset “Device select”
                stb     ADVCE,[BADDR]       ;start ADC conversion

; read ADC
                orb     PORT2,#80h          ;set “Device Select”
                ldb     ADVCE,#01h          ;select ADC
                stb     ADVCE,[BADDR]       ;write to “Device Select” register on S4580
                andb    PORT2,#7fh          ;reset “Device Select”
                ld      R0,BADDR            ;pointer to ADC low byte
                ld      R1,BADDR            ;pointer to ADC high byte
                inc     R1
                ldb     R2,#02h             ;wait for conversion to complete
AdcWait:        djnz    R2,AdcWait
                ldb     AINL,[R0]           ;read ADC low byte
                ldb     AINH,[R1]           ;read ADC high byte

; B100 ( AINL  )  Analog Input  (low)
; B101 ( AINH  )  Analog Input (High)
; B102 ( ASEL  )  Select Analog Input
; B103 (ADVCE)  Analog Device Select
; W104 (BADDR) S4580 Slot Address
```

User function 4:

****

block: 1 - Assembly

```
; This user function writes all analog output values to the S4580. Prior to calling this user function,
; "BADDR" must be loaded with the byte 0 slot address the S4580 is in (i.e. 7424 for slot 0-0, 7488 for
; slot 0-1, etc.). The values loaded into the AOUT0 thru AOUT3 analog output image registers must be
; in the range of 0 to fffH where 0 corresponds to -10V out, 800H corresponds to 0V out, and fffH corresponds
; to +10V out.
;
                .equ    ADVCE, B103
                .equ    BADDR,W104
                .equ    AOUT0,W120
                .equ    AOUT1,W122
                .equ    AOUT2,W124
                .equ    AOUT3,W126

; write AOUT0 and AOUT1 data
                orb     PORT2,#80h          ;set "Device Select"
                ldb     ADVCE,#04h          ;select "DAC0/1"
                stb     ADVCE,[BADDR]       ;write to "Device Select" register on S4580
                andb    PORT2,#7fh          ;reset "Device Select"

                ld      R0,BADDR            ;temporary pointer to S4580
                stb     AOUT0,[R0]+         ;write AOUT0L to DAC0
                stb     AOUT0+1,[R0]+       ;write AOUT0H to DAC0
                stb     AOUT1,[R0]+         ;write AOUT1L to DAC1
                stb     AOUT1+1,[R0]        ;write AOUT1H to DAC1

; write AOUT2 and AOUT3 data
                orb     PORT2,#80h          ;set "Device Select"
                ldb     ADVCE,#05h          ;select "DAC2/3"
                stb     ADVCE,[BADDR]       ;write to "Device Select" register on S4560
                andb    PORT2,#7fh          ;reset "Device Select"

                ld      R0,BADDR            ;temporary pointer to S4580
                stb     AOUT2,[R0]+         ;write AOUT2L to DAC2
                stb     AOUT2+1,[R0]+       ;write AOUT2H to DAC2
                stb     AOUT3,[R0]+         ;write AOUT3L to DAC3
                stb     AOUT3+1,[R0]        ;write AOUT3H to DAC3

; load (update) DACs
                orb     PORT2,#80h          ;set "Device Select"
                ldb     ADVCE,#02h          ;select  "Conversion Start"
                stb     ADVCE,[BADDR]       ;write to "Device Select" on S4580
                andb    PORT2,#7fh          ;reset "Device Select"
                ld      R0,BADDR            ;point to "Load DAC" register on S4580
                inc     R0
                stb     ADVCE,[R0]          ;update analog outputs (Load DACs)

; B103 (ADVCE)  Analog Device Select
; W104 (BADDR)  S4520 Slot Address
; W120 (AOUT0)  Analog Output0 value
; W122 (AOUT1)  Analog Output1 value
; W124 (AOUT2)  Analog Output2 value
; W126 (AOUT3)  Analog Output3 value
```

Installation

Prior to installing the S4580, the I/O slot addressing dip switch on the board must be set for the slot the board will be addressed as.

Note: Geographical addressing is not used in the M4500. The slot the S4580 is addressed as is solely defined by the dip switch settings on the S4580 itself not by the slot in the M4500 chassis that the board is placed in. Two poles on the dip switch of the board set the binary slot address of the board as follows:

S4580 SW1 Dip Switch Slot Addressing

2	1	Slot Address
off	off	0
off	on	1
on	off	2
on	on	3

The SW1 Slot address dip switch is located in the upper left hand corner of the component side of the S4580. The respective switch pole is "on" when in either the "on" or "close" position and "off" when either in the "off" or "open" position depending on the type of dip switch used.

To install the S4580 in the M4500 chassis, turn power to the M4500 "off" and remove the cover plate of the M4500 by loosening the captive screws that retain it. Install the S4580 in the respective slot of the M4500, making sure the DIN connector on the S4580 fully mates with the DIN connector in the M4500 motherboard and that the top of the S4580 is seated correctly in the card guides at the top of the M4500. Install the M4500 cover back onto the M4500 making sure the field wiring connector protrudes through the respective opening in the cover. The M4500 cover will retain the S4580 both from the top and the front, holding the S4580 in place during normal operation. Tighten the captive screws that retain the cover on the M4500. Install the female field wiring connector to the corresponding male connector on the S4580. The S4580 is now installed and ready to run. To remove the S4580, simply perform the previous steps in reverse.

Specifications

Board Size:

Length:	6.50”
Height:	4.25”
Width:	0.80”

Power Requirements (P4500):

+5VDC:	100 milliamps
+12VDC:	50 milliamps
-12VDC:	50 milliamps

Temperature Ranges:

Storage:	0 to 85 degrees C
Operating:	0 to 60 degrees C

Relative Humidity: 5 to 95%

Input Section:

Number of Analog Inputs:	4
Input Range:	-10V to +10V
DC Input Resistance:	1M ohms or greater
Maximum Input Voltage:	-30V to +30V
Accuracy:	0.5%
Repeatability:	0.1%
Analog-to-Digital Conversion Method:	Successive Approximation
Analog-to-Digital Conversion Time:	15 microseconds
Resolution:	14-bit

Specifications (cont'd)

Output Section:

Number of Analog Outputs:	4
Output Range:	-10V to +10V
Load Resistance:	2K ohms or greater
Accuracy:	0.5%
Repeatability:	0.75%
Digital-to-Analog Conversion time:	10 microseconds
Resolution:	12-bit
Short Circuit Duration:	continuous



Figure 1
Typical User Wiring



Figure 2
Board Outline